# LABORATORY 5: Canonical form of second order PDE with two variables

## Initialization

```
> restart:
> with(DEtools):
> with(PDEtools):
> with(plots):
Warning, the name changecoords has been redefined
```

## Canonical form

General form of second order quasilinear PDE with two variables is

(1)
$$a(x,y)\left(\frac{\partial^2}{\partial x^2}u(x,y)\right)+b(x,y)\left(\frac{\partial^2}{\partial y\,\partial x}u(x,y)\right)+c(x,y)\left(\frac{\partial^2}{\partial y^2}u(x,y)\right)+\Phi\left(x,y,u(x,y),\frac{\partial}{\partial x}u(x,y),\frac{\partial}{\partial y}u(x,y)\right)=0$$

Canonical form is obtained for $a, c \in \{-1, 0, 1\}$ and $b = 0$ .
In the case of second order quasilinear PDE with two variables there exists a linear change of variables such that equation (1) to be written in the canonical form.

Starting from the principal part of (1)

$$a(x,y)\left(\frac{\partial^2}{\partial x^2}u(x,y)\right)+b(x,y)\left(\frac{\partial^2}{\partial y\,\partial x}u(x,y)\right)+c(x,y)\left(\frac{\partial^2}{\partial y^2}u(x,y)\right)$$

we construct the characteristics equation:

$$a(x,y)\,dy^2-b(x,y)\,dx\,dy+c(x,y)\,dx^2=0$$

or

(2)
$$a(x,y)\left(\frac{dy}{dx}\right)^2-b(x,y)\frac{dy}{dx}+c(x,y)=0$$

The change of variables is given by the solutions of (2). The value of $\Delta$ is

$$\Delta=b^2-4\,a\,c$$

**I.** If $0<\Delta$ then the equation (1) is a **hyperbolic equation** and we have

$$\frac{dy}{dx}=\frac{b+\sqrt{\Delta}}{2\,a} \quad \text{and} \quad \frac{dy}{dx}=\frac{b-\sqrt{\Delta}}{2\,a}$$

Solving these differential equations we get the solutions

$$\phi(x,y)=c_1$$
$$\psi(x,y)=c_2$$

The change of variables in this case is

$$s=\phi(x,y)$$
$$t=\psi(x,y)$$

making this change of variables we get

$$\left(\frac{\partial^2}{\partial t\,\partial s}\mathrm{u}(s,t)\right)+\mathrm{F}\left(s,t,\mathrm{u}(s,t),\frac{\partial}{\partial s}\mathrm{u}(s,t),\frac{\partial}{\partial t}\mathrm{u}(s,t)\right)=0$$

which is called *first canonical form* of hyperbolic equation. Notice that this first canonical form is not a canonical form in the sense of the definition, but this form is important in solving this type of equation. In order to get the canonical form (in the sense of the definition) we have to do another change of variables, namely

$$\xi = s + t$$
$$\eta = s - t$$

Using this change of variables we obtain

$$\left(\frac{\partial^2}{\partial \xi^2}\mathrm{u}(\xi,\eta)\right)-\left(\frac{\partial^2}{\partial \eta^2}\mathrm{u}(\xi,\eta)\right)+F_1\left(\xi,\eta,\mathrm{u}(\xi,\eta),\frac{\partial}{\partial \xi}\mathrm{u}(\xi,\eta),\frac{\partial}{\partial \eta}\mathrm{u}(\xi,\eta)\right)=0$$

**II.** If $\Delta = 0$ then the equation (1) is a **parabolic equation** and we have

$$\frac{dy}{dx}=\frac{b}{2\,a}$$

Solving this equation we get the solution

$$\phi(x,y)=c$$

The change of variables in this case is

$$s=\phi(x,y)$$
$$t=x$$

making this change of variables we get the canonical form

$$\left(\frac{\partial^2}{\partial t^2}\mathrm{u}(s,t)\right)+\mathrm{G}\left(s,t,\mathrm{u}(s,t),\frac{\partial}{\partial s}\mathrm{u}(s,t),\frac{\partial}{\partial t}\mathrm{u}(s,t)\right)=0$$

**III.** If $\Delta < 0$ then the equation (1) is an **elliptic equation** and we have

$$\frac{dy}{dx}=\frac{b+I\sqrt{-\Delta}}{2\,a} \quad \text{and} \quad \frac{dy}{dx}=\frac{b-I\sqrt{-\Delta}}{2\,a}$$

Solving these equations we get

$$\phi(x,y)+I\,\psi(x,y)=c_1$$
$$\phi(x,y)-I\,\psi(x,y)=c_2$$

The change of variables in this case is

$$s=\phi(x,y)$$
$$t=\psi(x,y)$$

making this change of variables we get

$$\left(\frac{\partial^2}{\partial s^2}\mathrm{u}(s,t)\right)+\left(\frac{\partial^2}{\partial t^2}\mathrm{u}(s,t)\right)+\mathrm{H}\left(s,t,\mathrm{u}(s,t),\frac{\partial}{\partial s}\mathrm{u}(s,t),\frac{\partial}{\partial t}\mathrm{u}(s,t)\right)=0$$

Let's consider the following PDE:

```
> PDE := diff(u(x,y),x,x)+5*diff(u(x,y),x,y)+6*diff(u(x,y),y,y)=0;
```

$$PDE := \left(\frac{\partial^2}{\partial x^2} u(x,y)\right) + 5\left(\frac{\partial^2}{\partial y\,\partial x} u(x,y)\right) + 6\left(\frac{\partial^2}{\partial y^2} u(x,y)\right) = 0$$

We intend to obtain the canonical form. First we identify the coefficients from the principal part, in this case are $a=1$ , $b=5$ and $c=6$ . Dorim sa aducem aceasta ecuatie la forma canonica. Pentru aceasta vom identifica coeficientii A,B,C.

```
> a:=1;
```

$$a := 1$$

```
> b:=5;
```

$$b := 5$$

```
> c:=6;
```

$$c := 6$$

Now we construct and we find the roots of the characteristics equation:

```
> Delta:=b^2-4*a*c;
```

$$\Delta := 1$$

So, $0 < \Delta$ then the equation is a hyperbolic equation

```
> ch_eq:=a*t^2-b*t+c=0;
```

$$ch\_eq := t^2 - 5t + 6 = 0$$

```
> s1:=solve(ch_eq,t);
```

$$s1 := 3, 2$$

```
> d_eq1:=diff(y(x),x)=s1[1];
```

$$d\_eq1 := \frac{d}{dx} y(x) = 3$$

```
> d_eq2:=diff(y(x),x)=s1[2];
```

$$d\_eq2 := \frac{d}{dx} y(x) = 2$$

We solve these differential equations

```
> f1:=dsolve(d_eq1);
```

$$f1 := y(x) = 3x + \_C1$$

```
> f2:=dsolve(d_eq2);
```

$$f2 := y(x) = 2x + \_C1$$

and we construct the new variables expressions.

```
> expr1:=s=subs(y(x)=y,solve(f1,_C1));
```

$$expr1 := s = y - 3x$$

```
> expr2:=t=subs(y(x)=y,solve(f2,_C1));
```

$$expr2 := t = y - 2x$$

From **expr1** and **expr2** we obtain $s$ and $t$ with respect to $x$ and $y$

```
> tr:=solve({expr1,expr2},{x,y});
```

$$tr := \{y = 3\,t - 2\,s,\ x = -s + t\}$$

The change of variables can be made in MAPLE using **dchange** command

```
> PDE1:=dchange(tr,PDE);
```

$$PDE1 := -\left(\frac{\partial^2}{\partial t\,\partial s}\,\mathrm{u}(s, t)\right) = 0$$

Since our equation is a hyperbolic one we get the first canonical form. To get the canonical form (in the definition sense) we have to do a second change of variables:
Now we apply the same steps as in the case of the first canonical form

```
> trr:=solve ({xi=s+t,eta=s-t},{s,t});
```

$$trr := \{t = -\frac{\eta}{2} + \frac{\xi}{2},\ s = \frac{\xi}{2} + \frac{\eta}{2}\}$$

```
> dchange(trr,PDE1);
```

$$\left(\frac{\partial^2}{\partial \eta^2}\,\mathrm{u}(\eta, \xi)\right) - \left(\frac{\partial^2}{\partial \xi^2}\,\mathrm{u}(\eta, \xi)\right) = 0$$

To find the solution of the initial PDE we use the **pdsolve** command:

```
> pdsolve(PDE);
```

$$\mathrm{u}(x, y) = \_F1(y - 2\,x) + \_F2(y - 3\,x)$$

```
> restart:
>
> with(DEtools):with(PDEtools):
```

Let's cosider the following PDE:

```
> PDE := diff(u(x,y),x,x)+diff(u(x,y),x,y)+diff(u(x,y),y,y)=0;
```

$$PDE := \left(\frac{\partial^2}{\partial x^2}\,\mathrm{u}(x, y)\right) + \left(\frac{\partial^2}{\partial y\,\partial x}\,\mathrm{u}(x, y)\right) + \left(\frac{\partial^2}{\partial y^2}\,\mathrm{u}(x, y)\right) = 0$$

We initialize the coefficients

```
> a:=1;
```

$$a := 1$$

```
> b:=1;
```

$$b := 1$$

```
> c:=1;
```

$$c := 1$$

We evaluate $\Delta$ to find the type of the equation

```
> Delta:=b^2-4*a*c;
```

$$\Delta := -3$$

In this case $\Delta < 0$ so this equation is an elliptic equation.

```
> ch_eq:=a*t^2-b*t+c=0;
```

$$ch\_eq := t^2 - t + 1 = 0$$

```
> s1:=solve(ch_eq,t);
```

$$s1 := \frac{1}{2}+\frac{1}{2}I\sqrt{3}, \frac{1}{2}-\frac{1}{2}I\sqrt{3}$$

We choose one solution and we consider the corresponding differential equation

```
> d_eq1:=diff(y(x),x)=s1[1];
```

$$d\_eq1 := \frac{d}{dx}\mathrm{y}(x) = \frac{1}{2}+\frac{1}{2}I\sqrt{3}$$

```
> f1:=dsolve(d_eq1);
```

$$f1 := \mathrm{y}(x) = \frac{x}{2}+\frac{1}{2}I\,x\sqrt{3}+\_C1$$

We find the expression of the first integral

```
> expr:=subs(y(x)=y,solve(f1,_C1));
```

$$expr := y-\frac{x}{2}-\frac{1}{2}I\,x\sqrt{3}$$

In the case of elliptic equation the change of variables is given by the real part and the imaginary part of the expression. To split the real part and imaginary part of the above expression we have to use the functions **`Re(x)`** and **`Im(x)`** combined with the **`evalc`** command

```
> Re(expr);Im(expr);
```

$$\frac{1}{2}\Re(2\,y-x-x\sqrt{3}\,I)$$

$$\frac{1}{2}\Im(2\,y-x-x\sqrt{3}\,I)$$

```
> evalc(Re(expr));evalc(Im(expr));
```

$$y-\frac{x}{2}$$

$$-\frac{x\sqrt{3}}{2}$$

```
> expr1:=s=evalc(Re(expr));
```

$$expr1 := s = y-\frac{x}{2}$$

```
> expr2:=t=evalc(Im(expr));
```

$$expr2 := t = -\frac{x\sqrt{3}}{2}$$

\>

Now, we find x and y with respt to s and t in order to make the change of variables.

```
> tr:=solve({expr1,expr2},{x,y});
```

$$tr := \{y = -\frac{\sqrt{3}\,t}{3}+s, x = -\frac{2\sqrt{3}\,t}{3}\}$$

Using **`dchange`** command we make the change of variables:.

```
> PDE1:=dchange(tr,PDE);
```

$$PDE1 := \frac{3}{4}\left(\frac{\partial^2}{\partial s^2}\mathrm{u}(s,t)\right) - \frac{1}{4}\left(\frac{\partial^2}{\partial t\,\partial s}\mathrm{u}(s,t)\right)\sqrt{3} - \frac{1}{2}\left(-\frac{1}{2}\left(\frac{\partial^2}{\partial t\,\partial s}\mathrm{u}(s,t)\right) - \frac{1}{2}\left(\frac{\partial^2}{\partial t^2}\mathrm{u}(s,t)\right)\sqrt{3}\right)\sqrt{3} = 0$$

.

```
> simplify(%);
```

$$\frac{3}{4}\left(\frac{\partial^2}{\partial s^2}\mathrm{u}(s,t)\right) + \frac{3}{4}\left(\frac{\partial^2}{\partial t^2}\mathrm{u}(s,t)\right) = 0$$